
Fusebox 101 

Who needs Fusebox? What's the big fuss about an application's architecture? After all, applications aren't essentially about architecture--they're about doing things--things like logging in, querying and displaying a result set from a database, or making a request for a credit card validation. 

Suppose that a user wishes to log into an application. The user enters their username and password and clicks a submit button. That can be handled by writing a file called Login.cfm, whose code looks like this:

<form action="ValidateUser.cfm" method="post">

   User Name: <input type="Text" name="userName"><br>

   Password: <input type="Password" name="password"><br>

   <input type="Submit" value=" ok ">

</form>

Another file, ValidateUser.cfm, then would be written to accept the form variables passed to it, run a query against a database and then either show the user a menu or return them to try another login. 

<cfquery datasource="myDSN" name="myQ">

   SELECT userID, firstName, lastName

   FROM Users 

   WHERE userName = '#form.userName#'

   AND password = '#form.password#'

</cfquery>

<cfif myQ.recordCount>

   <cflocation url="UserMenu.cfm">

<cfelse>

   <cflocation url="Login.cfm">

</cfif>

What could be simpler? 

Not much--and if your applications aren't any more complex than this, you won't need Fusebox--or any other methodology. Unfortunately, applications are never this simple. Even applications that start off simple--where the client promises that they will stay simple--almost never do. I've become convinced that the most dangerous words a client can utter are, "Wouldn't it be nice..." 

In fact, studies show that between 70% and 90% of the entire cost of an application over its life is the ongoing maintenance of the application. It's a universal truism that developers would almost always rather rewrite than try to maintain someone else's code. Why is this? One excellent reason is that in a real-world application (unlike the simplistic one I've shown), the direct links from one code file to another--all those <cflocation>s and <form action>s--become enormously difficult to keep track of. 

Whether maintaining an application or developing it fresh, another problem rears itself--if you make a change in one file, what others files will be effected? The problem is even worse if several developers are working on the same application. The bigger the application, the more there is to keep track of--until the complexity and the interconnections preclude any but the most experienced developers from working on it. If you think of a pyramid turned upside down--and you at the bottom like Atlas carrying the world atop his shoulders, you won't be far off. And heaven help you if you have a distributed development team. The problems multiply geometrically and we have tangled code that looks like a plate of spaghetti. 

I won't belabor the difficulties of development as you probably wouldn't be looking at Fusebox if you hadn't experienced either these or other problems. The point I'd like to make is that the problems come about as a natural result of the normal development process. There's nothing mysterious about it; there aren't evil forces conspiring to make your life difficult, no matter how true that may seem at 2 am of the night before a deadline. If you want to change the results, you must change the process.

So how does Fusebox help--and how much is this help going to cost in terms of changing the way you write code? 

Surprisingly little. Here is that same code you saw a minute ago written to the Fusebox methodology: 

<form 

   action="index.cfm?fuseaction="#attributes.XFA.validateUser#" 

   method="post">

   User Name: <input type="Text" name="userName"><br>

   Password: <input type="Password" name="password"><br>

   <input type="Submit" value=" ok ">

</form>
Fusebox helps developers build robust and scalable web applications easily, surely, and quickly--and it's surprisingly simple. The tangle of spaghetti code is replaced with a simple, ordered, easy-to-maintain architecture: 

A Fusebox application works by responding to requests to do something--a fuseaction, in Fusebox parlance. This request may come about as a result of a user action--a user submitting a form, for example, or clicking a link--or it may occur as a result of a system request. Let's consider the first situation. 

You type in the URL for www.halhelms.com, your favorite Fusebox site! You press the secret combination of keys while doing the prescribed breathing exercises that gets you to the secret portion of the site. There you are asked for your username and password. This is the code you saw on the last page: 

<form 

   action="index.cfm?fuseaction="#attributes.XFA.validateUser#" 

   method="post">

   User Name: <input type="Text" name="userName"><br>

   Password: <input type="Password" name="password"><br>

   <input type="Submit" value=" ok ">

</form>

When you submit the form, index.cfm is called. In fact, everything an application can do is done by sending a fuseaction request to index.cfm. This file, so central to the methodology, is called the fusebox. When a fusebox is called, a variable called "fuseaction" is also sent. 

The fusebox's main job is to route a fuseaction request to one or more code files called fuses. These files are typically small and have well-defined roles.

Here is the code for a really small fusebox--one that is only capable of handling a couple of different fuseactions. 

<cf_formURL2attributes>

<!--- In case no fuseaction was provided, I'll set up one 

to use by default. --->

<cfparam name="attributes.fuseaction" default="login">

<cfinclude template="myGlobals.cfm">

<cfswitch expression = "#attributes.fuseaction#">

<cfcase value="login">

   <cfset attributes.XFA.submitForm = "validateLogin">

   <cfinclude template="dspLogin.cfm">

</cfcase>

<cfcase value="validateLogin">

   <cfset attributes.XFA.successfulLogin = "showUserMenu">

   <cfset attributes.XFA.failedLogin = "login&badLogin=True">

   <cfinclude template="actValidateUser.cfm">

</cfcase>

</cfswitch>
The first part of the Fusebox takes care of some housekeeping chores that I'll discuss later. For now, I want to draw your attention to the navy color-coded section, where the routing is being done. 

The routing begins with a <cfswitch> statement that examines the value of attributes.fuseaction. 

Why attributes.fuseaction you might ask? 

The technical reason is that the custom tag called above, <cf_formURL2attributes>, converts all form and URL variables into attributes-style variables. The larger reason is--well, I'd like to defer the larger reason for a while. So far, you know that a fusebox is looking for an attributes-scoped variable called fuseaction. 

Once it finds a matching value in one of the <cfcase> statements, the code between those particular <cfcase> tags is executed. 

<cfcase value="login">

   <cfset attributes.XFA.submitForm = "validateLogin">

   <cfinclude template="dspLogin.cfm">

</cfcase>

In the Fusebox methodology, the <cfcase> code is used to set up an environment in which one or more fuses can be called that will perform whatever actions are needed to do the work requested by the variable fuseaction. 

What sort of things might a fuse do? Things like display a form, see if a user's password and username match those found in a database, show a menu of user options--in short, anything a web application can do will be done through the use of fuses. Earlier, I mentioned that well-written fuses are very short, restricting themselves to doing only one or two things. Why? Well, short fuses are easier to write (and maintain), are less buggy and easier to debug, and facilitate code reuse. After all, the odds of you needing one "do-it-all" fuse in several different places are much less than those of you reusing code that only handles a single problem. 

Fuses have one or more exit points--areas where the action will return to the fusebox. For example, clicking the submit button on a form is an exit point--when the user clicks "Submit", the action goes back to the fusebox (as you saw on a previous page). Every link on a user menu is an exit point (as each click will cause the browser to return to the fusebox), each drilldown to get more information is an exit point. 

Some exit points are visible and require user interaction, such as submitting a form or clicking a link; other exit points are not visible. For example, look at this fuse that validates a user's login:

<cfquery datasource="#request.dsn#" name="FindUser">

   SELECT userID, firstName, lastName, userGroups

   FROM Users

   WHERE userName = '#attributes.userName#' 

   AND password = '#attributes.password#'

</cfquery>

<cfif FindUser.recordcount>

   <cflocation url="index.cfm?fuseaction=#attributes.XFA.successfulValidation#">

<cfelse>

   <cflocation url="index.cfm?fuseaction=#attributes.XFA.failedValidation#">

</cfif>

Here, we have two different exit points, neither of which require user interaction. If the results of a query return any rows--that is, if the query found a user--the action returns to the fusebox with a fuseaction having the value of attributes.XFA.successfulValidation, while failure to find a user will result in returning to the fusebox with a fuseaction having the value of attributes.XFA.failedValidation. 

Whether generated by user interaction or by the application itself, exit points in fuses are always return to the fusebox with a fuseaction. 

Can you identify the exit points in this fuse? 

<cftry>

<cflock scope="session" timeout="10" throwontimeout="Yes">

   <cfset myCurrentUser = session.currentUser>

</cflock>

<cfcatch>

   <cflocation url="index.cfm?fuseaction=#attributes.XFA.retryUser#">

</cfcatch> 

</cftry>

<cfif myCurrentUser.booksRated LT 100>

   I'm sorry, but this area is only for Premier Book Partners. 

   Click <a href="index.cfm?fuseaction=#attributes.XFA.mainMenu#">

   here</a> to go back to the main menu.

<cfelse>

   Welcome to the Premier Book Partners site. Are you interested in taking a

   quick tour?

   <form 

      action="index.cfm?fuseaction=#attributes.XFA.submitForm#" 

      method="post">

      <input 

         type="Submit" 

         name="yes" 

         value="Yes"> 

      <input 

         type="Submit" 

         name="no" 

         value="No">

   </form>

</cfif>

Here's that same code--only this time I've color-coded each exit point in green. 

<cftry>

<cflock scope="session" timeout="10" throwontimeout="Yes">

   <cfset myCurrentUser = session.currentUser>

</cflock>

<cfcatch>
   <cflocation url="index.cfm?fuseaction=#attributes.XFA.retryUser#">
</cfcatch> 

</cftry>

<cfif myCurrentUser.booksRated LT 100>

   I'm sorry, but this area is only for Premier Book Partners. 

   Click <a href="index.cfm?fuseaction=#attributes.XFA.mainMenu#">

   here</a> to go back to the main menu.

<cfelse>

   Welcome to the Premier Book Partners site. Are you interested in taking a

   quick tour?

   <form 
      action="index.cfm?fuseaction=#attributes.XFA.submitForm#" 

      method="post">

      <input 

         type="Submit" 

         name="yes" 

         value="Yes"> 

      <input 

         type="Submit" 

         name="no" 

         value="No">

   </form>

</cfif>

Notice that each exit point has a fuseaction associated with it that begins with an "XFA." This is shorthand for "eXit FuseAction". You could just hardcode a fuseaction at every exit point--something like this: 

Click <a href="index.cfm?fuseaction=mainMenu">

   here</a> to go back to the main menu.

In fact, many Fusebox writers and coders do just that. But this impairs code reusability greatly. If you want to reuse a fuse--whether in another application or in a different point in the same one--you now must deal with the fact that your exit fuseactions may not be the same, varying according to the context in which they are used. This means that you must begin introducing conditional statements in your code: 

<form action="index.cfm?fuseaction=<cfif attributes.fuseaction 

is "newUser">insertUser<cfelseif attributes.fuseaction is

"editUser">updateUser</cfif>" method="post">

Now, each time you wish to reuse the fuse, you must open up the file, alter the existing code and save it, introducing the very real possibility that you will introduce a bug into the code. Further, it reduces readability (and maintainability) of the code, and makes it just plain ugly.

To me, though, the worst thing about this practice is that it requires the coder of one fuse know too much about the entire application. One of the great things about Fusebox is that it can free us developers from the "inverted pyramid" syndrome I spoke of earlier. Using XFAs lets us create fuses that will be used anywhere--without worrying about where a form should be submitted to or a link pointed to. You create variables for these exit fuseactions and then let the application's fusebox set the values for these at run time: 

<cfcase value="validateLogin">

   <cfset attributes.XFA.successfulValidation = "mainMenu">

   <cfset attributes.XFA.failedValidation = "login&badLogin=True">

   <cfinclude template="actValidateUser.cfm">

</cfcase>

A recap of this section: 

The program flow of a Fusebox application: 

1. A fuseaction request is sent to the fusebox. 
2. The fusebox calls one or more fuses to perform some actions related to this request.
3. The fuse(s) themselves contain exit fuseactions, 
4. One of which will be sent back to the fusebox 

and the process repeats itself. 

It's helpful to categorize fuses into different types. For example, some fuses display information, forms, etc. to users; others work behind the scenes to do things like process credit cards; still others are responsible for querying databases. Many Fusebox developers find it helpful to use a prefix when naming a fuse that conveys the fuse type. Examples of these are: 

	dspShowProductInfo.cfm
	a display type fuse used to show or request information from a user

	act_saveUserInfo.cfm
	an action type fuse used to perform an action without displaying information to a user

	qryGetUserInfo.cfm
	a query type fuse used to interact with data sources without displaying information to a user

	url_processOrder.cfm
	an action type fuse used to redirect an action???

	app_Locals.cfm
	an application type fuse used to set up the environment in which the application runs


These naming conventions are best thought of as suggestions: You can use the naming conventions outlined here--or not, depending on what suits you best. 

If you do use them, will you use an underscore to separate prefix from fusename (dsp_fusename.cfm)--or will you use the mixed case spelling that many developers prefer (actAnotherFuseName.cfm)? 

I like the mixed case usage, but you may prefer something else. Whatever you choose, don't get bogged down in disputes over naming conventions; Fusebox is primarily about developing successful applications, not about naming schemes. 

You have seen how fuseactions are returned to the fusebox, and you have had a glimpse of the mechanism the fusebox uses to call helper fuses (a <cfswitch> statement). Fuses really do all the work in a Fusebox application. The fusebox itself acts like a manager, delegating work to one or more of these fuses. 

One of the principles of modern programming is encapsulation, which states that, as much as is possible and reasonable, applications should be divided into areas of related functionality. We do this constantly in many aspects of our lives. Animal bodies are made up of various systems: circulatory, respiratory, skeletal, etc.. Companies are composed of various functions: human resources, sales, production, etc.. 

Applying this principle to the Fusebox methodology, we get the concept of circuit applications. Instead one single, enormous Fusebox application (with a monolithic fusebox), we can now break a complex application into several (hopefully simpler) parts--each with their own fusebox. Together, these constitute the entire application, which is called the home application. 

Circuit applications are groupings of fuses that, at least in the Fusebox architect's mind, makes logical and coherent sense. You commonly see circuit applications for things like a shopping cart, or an admin module, or a user module. 

Each one of these circuit applications is a little self-contained Fusebox application. In order to make them work together, the original Fusebox specification directs that an app_Locals.cfm file should be created for each circuit application and an app_Globals.cfm file for the home application. 

The job of these "app" files is to set up the environment in which the application will run. So, things that are application-global--an application datasource name, for example--will be set in app_Globals.cfm and variables specific to a circuit application are set in the app_Locals.cfm file. 

There is only one app_Globals file, set in the home application's directory. Here is an example app_Globals.cfm file: 

<cfapplication 

   name="MyAppName" 

   clientmanagement="Yes" 

   clientstorage="cookie">

<cfset request.mainDSN = "MyDSN">

<cflock type="EXCLUSIVE" scope="APPLICATION" timeout="5">

   <cfif NOT IsDefined( 'application.EmployeeList' )>

      <cfinclude template="queries/qryEmployeeList.cfm">

   </cfif>

</cflock>
Some "environment variables" pertain only to the particular circuit application; these should be stored in the app_Locals.cfm file. Each circuit application can define its own app_Locals.cfm file. 

<cfset userDSN = "Users">

<cfset userStyleSheet = "UserStyle.css">
One of the great things about Fusebox is the flexibility it gives developers. This not only lets you decide on things like naming conventions, but also allows Fusebox developers to experiment with new ideas without fear of running afoul of any Fusebox police. This, in turn, lets Fusebox evolve, whether to stay abreast of technology developments or simply to incorporate good ideas that weren't originally envisioned. Different developers have different goals and bring with them different techniques. Such creative diversity can only be good for the methodology. 

One idea that has proven very successful in the object-oriented world is that of inheritance, a mechanism for reusing code (surely the Holy Grail of many programmers). Having written a perfectly good circuit application--say a user module--for one application, we surely want to use it in others, and would like to do so without having to make wholesale changes to the code. 

The original Fusebox specification doesn't make this particularly easy, so you'll have to muck about, changing the app_Locals.cfm and the app_Globals.cfm. I think this idea of inheritance can make this process both easier and safer. 

Instead of having separate fuseboxes independent of each other, I use the concept of nested fuseboxes. 

The home application itself is a fusebox that may delegate work to a fuse--or to a nested fusebox. Under this scheme, the information in app_Globals and app_Locals is combined into a single myGlobals.cfm file that works whether operating in stand-alone mode (as a home application) or if nested in the context of a larger application. 

Code for myGlobals.cfm must recognize that the application may be operating as either a home application or it may have been incorporated as a circuit application. If I'm a home app, I will want to set up state management with <cfapplication>, but I don't want to do so if it's already been set up. I use the presence of application.applicationName to indicate which mode myGlobals.cfm is in. 

The logic goes like this: I am a myGlobals file, but I don't know if I'm the home app or if someone is using me as a circuit application. However, I do know that if application.applicationName already exists, it will have been set by my a preceding fusebox using the <cfapplication name="App_Name"> tag. That means that I'm not the home application, and I won't set up state management with <cfapplication>. 

Here's an example of myGlobals.cfm: 

<cfif NOT IsDefined( 'application.applicationName' )>

   <cfapplication 

      clientmanagement="Yes" 

      clientstorage="cookie" 

      name="MyAppName">

</cfif>
Other variables can be set using <cfparam> to make sure that variables previously set aren't unintentionally overwritten. 

<cfparam name="request.mainDSN" default="MyDSN">

<cfparam name="request.mainStyleSheet" default="StyleA.css">
Each nested fusebox inherits the environment from its "parent", overwriting particular varibles if need be. That seems to me a much simpler mechanism and simplicity, for me at least, is the true Holy Grail of programming. 

Having created the idea of nesting fuseboxes two levels deep, can we go deeper? Yes--and though I don't often find a need for this, I have done it in some particularly complex applications--and was glad I had the ability to do so. 

There are a couple of other excellent reasons for nesting fuseboxes. One concerns security and authentication issues, in which the ability to inherit permissions aids greatly in creating a coherent security scheme. The other is in exception and error handling; there applications should have a common method for dealing with global errors and exceptions. Inheritance through nested fuseboxes lets exceptions "bubble up" to parent applications, letting circuit applications handle only those type of exceptions that are specific to the circuit. [See the lessons on Security/Authentication and Error/Exception Handling at www.halhelms.com for more on these.]

With all that theory behind us, let's take a closer look at the fusebox itself--one that works with the nested fusebox scheme. Here's a skeleton I use when beginning to create the fusebox: 

<!-- index.cfm -->

<!--- Part of the Fusebox method is to convert incoming form 

or URL variables into attributes variables. 

This custom tag does the trick. --->

<cf_formURL2attributes>

<!--- In case no fuseaction was given, I'll set up one 

to use by default. --->

<cfparam name="attributes.fuseaction" default="bas.1a">

<!--- Am I the home application? --->

<cfif NOT IsDefined( 'application.applicationName' )>

   <cfapplication clientmanagement="Yes" name="Application_Name">

</cfif>   

<!--- I can't count on application.cfm firing in case I get 

called as a custom tag, so I have to have my own way of a 

"universal include". --->

<cfinclude template="myGlobals.cfm">

<!--- Include Circuits if I'm the home app--->

<cftry>

   <cfif NOT IsDefined( 'request.Circuits' )>

      <cfinclude template="Circuits.cfm"> 

   </cfif>

   <cfcatch type="MissingInclude">

      I couldn't find the file Circuits.cfm.

      <cfabort>

   </cfcatch>

</cftry>

<!--- Translate the fuseaction sent to me into a fully qualified 

fuseaction --->

<cfparam 

   name="faTranslated" 

   default="FALSE" 

   type="boolean">

<cftry>

   <cfif NOT faTranslated AND ListLen( attributes.fuseaction, '.' ) GT 1>

      <cfset attributes.fuseaction = 

      request.circuits[ListFirst( attributes.fuseaction, '.' )] & '.' 

      & ListLast( attributes.fuseaction, '.' )>

      <cfset faTranslated = TRUE>

   </cfif>

   <cfcatch>

      I could not resolve the circuit application prefix sent to me.

      <cfabort>

   </cfcatch>

</cftry>

<cfif ListLen( attributes.fuseaction, '.' ) GT 1>

   <cfset attributes.fuseaction = 

   ListDeleteAt( attributes.fuseaction, 1, '.' )>

</cfif>

<!--- Peel off the first layer and determine whether to delegate 

the fuseaction. --->

<cfif ListLen( attributes.fuseaction, '.' ) GT 1>

   <cfinclude template = 

   "#ListFirst( attributes.fuseaction, '.' )#/index.cfm">

<cfelse>

<!--- I'm going to get one (and only one) fuseaction at a time.

Fuseactions correspond to methods or messages in OO languages. --->

   <cfswitch expression = "#attributes.fuseaction#">

[image: image17.png]


 <cfdefaultcase>

   I received a fuseaction called <B><FONT COLOR="000066">

"#attributes.fuseaction#"</FONT></B> that I don't have 

a handler for.

      </cfdefaultcase>

   </cfswitch>

</cfif>
Let's break the fusebox down... 

Let's walk through the fusebox, so you can see what each part does. 

<!--- Part of the Fusebox method is to convert incoming form 

or URL variables into attributes variables. 

This custom tag does the trick. --->

<cf_formURL2attributes>

I said earlier I wanted to defer the question about the use of attributes variables. Well, it's time to come clean. Here, we're face to face with a call to a custom tag that converts form and URL variables sent into the fusebox into attributes type variables. The question is, Why bother? 

The "conventional wisdom" (if such a thing can be said to exist with so new a methdology) is that the attributes scope is needed because of the two ways fuses--and circuit apps themselves--can be called. As you know, when a fuseaction is sent to the fusebox, the fusebox uses a <cfswitch> statement to determine which fuses to call on. The examples I've shown you so far use <cfinclude>s to call the fuse: 

<cfcase value="showSpecialPromotion">

   <cfset attributes.XFA.addItemToCart = "crt.addToCart">

   <cfset attributes.XFA.mainMenu = "bbh.home">

   <cfset attributes.productName = "Megadeveloper 2000">

   <cfinclude template="dspProductDetail.cfm">

</cfcase>
Although I suggest you use this method, some people may wish to call the fuse as a custom tag: 

<cfcase value="showProductDetail">

   <cfmodule 

      template="dspProductDetail.cfm" 

      XFA.addItemToCart = "crt.addToCart" 

      XFA.mainMenu = "home"

      productName = "Megadeveloper 2000">

</cfcase>
If you've worked with custom tags at all, you know that they exist in a separate memory space from the page that called them. If PageA.cfm calls PageB.cfm as a custom tag and each one sets a variable called myVar, both variables can exist independently. Changes to one will not affect the other. This is done deliberately to prevent a custom tag from "stepping on" the calling page's variables. If you wish to make variables available to the custom tag, you should pass them explicitly (as the example above does). 

Within the custom tag, you refer to variables that have been explicitly passed in with the attributes. prefix. For example, if the custom tag needed to refer to the variable productName, it would do so like this:

Well, this month we're featuring a special on #attributes.productName#.
This means that a fuse (and, indeed, an entire circuit application) can be called in two modes: as an included page and as a custom tag. Now the conventional wisdom had it that if code had to work both as an include AND a custom tag, it would need to reference variables differently. I say this is conventional wisdom because it turns out not to be accurate. Custom tags can see form and URL variables just fine.

Still, I think a good argument can be made for keeping FormURL2Attributes for an entirely different reason, one that addresses the idea of distributed development. If you have taken the online class on Fusedocs, you're aware of how much stress I lay on the ability to divide an application into small pieces that can be distributed to a wide range of developers. Fusedoc makes this work by telling each developer what s/he is responsible for. 

For example this part of Fusedoc tells the developer to pass a variable called myVar (which will be a Boolean value) back to the fusebox: 

<-- myVar: a BOOLEAN
It does not tell the developer how to pass that--whether as a form or URL variable. This level of detail should be left to the coder to determine. But if we don't have a way of "homogenizing" variables into a common form, the coder of another page that will receive these variables will need to deal with the uncertainty of what kind of variable s/he will get. Sure, there are ways of doing this, but arguably the best is letting FormURL2Attributes handle this for you. 

You know that upon exiting a fuse, a fuseaction is normally sent back to the fusebox via code such as... 

<cflocation url="index.cfm?fuseaction=#attributes.XFA.continue#">
or 

<form action="index.cfm" method="post">

   <input type="Hidden" name="fuseaction" value="#attributes.XFA.submitForm#">

   ...

</form>
This works fine while the calls to the fusebox are being set by the program itself (as these two examples are). But for users to get to your application, they need to enter a URL, and they prefer simple URLs like www.halhelms.com. Since you don't want the code to break, you need a create a default fuseaction for when none is specified. You could use the <cfdefaultcase> for this, but <cfdefaultcase> is so helpful when used as a debugging tool (as I'll explain at section [image: image19.png]


) that I instead create a default fuseaction explicitly: 

<!--- In case no fuseaction was given, I'll set up one 

to use by default. --->

<cfparam name="attributes.fuseaction" default="home">
The next several sections ( [image: image21.png]

) are used to accomodate the nested fusebox architecture I discussed earlier. This code will also work fine if you choose not to nest fuseboxes. I won't bother with what each line is doing, but I do need to explain some essentials for using nested fuseboxes. 

If you choose to use nested fuseboxes, each "child" circuit should be physically located under their parent directory. Here is an example: 

The BooksByHal home application has a child called Catalog that has a child called Adults that, in turn, has a child called LocalInventory. 

One of the reasons I stated for preferring an inheritance scheme with Fusebox is that it allows you to start with a home application used for one purpose and then incorporate this (as a circuit application) into a larger application. In keeping with this goal, I did not want to force the name of each fuseaction for an application to be unique, as how could I guarantee that the coder of one circuit application would not use a fuseaction name that turned out to be the same as that used in another circuit? 

Obviously, though, we need some way of uniquely identifying fuseactions; else, how will the home application's fusebox know that the link to home means the home fuseaction of the shopping cart circuit and not the home fuseaction of the user circuit? And, in keeping with the stated goal, we want to let applications transition from stand-alone to circuit status with a minimum of fuss. 

We can accomplish this by defining uniqueness as the combination of circuit application and fuseaction. There may be many "home" (or whatever) fuseactions in a large application, but there can be only one in each circuit application, and those circuit applications must themselves have unique names. 

To identify a particular fuseaction, you can reference the circuit application name and the fuseaction name. To cut down on typing, I allow for an abbreviation of circuit application names, so that Shopping Cart may become crt or user module becomes usr. These "compound" fuseaction names are then set by the fusebox as exit fuseactions. Here are the abbreviations I decided on for the BooksByHal application 

	Directory Name 
	Abbreviation 

	BooksByHal 
	bbh 

	Catalog 
	cat 

	Adults 
	adu 

	AffiliateInventory 
	afl 

	LocalInventory 
	loc 

	... 


Here's an example of setting exit fuseactions in the fusebox: 

<cfcase value="displayPage">

   <cfset attributes.XFA.moreInfo = "cat.displayItem">

   <cfset attributes.XFA.checkAvailability = "loc.checkInventory">

   <cfinclude template = "dspDisplayPage.cfm">

</cfcase>
Using nested fuseboxes, all action returns to the home application--it is this fusebox that needs to know how to resolve fuseactions like cat.displayPage. In theory, the home app could "walk" the directory tree, trying each of its children (and its children's children and so forth) for the appropriate circuit application. This would certainly allow us the greatest flexibility, but would impose a severe performance penalty. 

A good solution is to "register" each circuit application with the home application. Then, the home application can use this information directly without walking a directory tree. 

Begin by creating a Circuits.cfm file in the directory of the home application that creates a request structure called Circuits. This structure should have a key for each application--both home and circuit--defining its "path": 

<cfset request.Circuits = StructNew()>

<cfset request.Circuits.bbh = "BooksByHal">

<cfset request.Circuits.cat = "BooksByHal.Catalog">

<cfset request.Circuits.kid = "BooksByHal.Catalog.Kids">

<cfset request.Circuits.adu = "BooksByHal.Catalog.Adults">

<cfset request.Circuits.afl = "BooksByHal.Catalog.Adults.AffiliateInventory">

<cfset request.Circuits.loc = "BooksByHal.Catalog.Adults.LocalInventory">
The code in section [image: image24.png]


is put in every fusebox. If the variable request.Circuits doesn't exist, it will include the Circuits.cfm file. 

<!--- Include Circuits if I'm the home app--->

<cftry>

   <cfif NOT IsDefined( 'request.Circuits' )>

      <cfinclude template="Circuits.cfm"> 

   </cfif>

   <cfcatch type="MissingInclude">

      I couldn't find the file Circuits.cfm.

      <cfabort>

   </cfcatch>

</cftry>
The effect of all this code is somewhat like peeling an onion. The onion, in this case, is the "fully qualified" fuseaction that's been processed, replacing the circuit abbreviation with the path found in request.Circuits. 

Section [image: image26.png]


determines whether the fuseaction belongs to this fusebox ("this" being whatever fusebox we're in at the moment). 

<!--- Peel off the first layer and determine whether to delegate 

the fuseaction. --->

<cfif ListLen( attributes.fuseaction, '.' ) GT 1>

   <cfinclude template = 

   "#ListFirst( attributes.fuseaction, '.' )#/index.cfm">

<cfelse>
If the fuseaction is meant for a further nested fusebox, it includes this fusebox; otherwise, it uses a <cfswitch> statement to call the fuse(s) needed to perform the action. 

<!--- I'm going to get one (and only one) fuseaction at a time.

Fuseactions correspond to methods or messages in OO languages. --->

   <cfswitch expression = "#attributes.fuseaction#">
Finally, if no <cfcase> value was found for attributes.fuseaction, the fusebox tells the user the name of the fuseaction that couldn't be handled. 

 <cfdefaultcase>

   I received a fuseaction called <B><FONT COLOR="000066">

"#attributes.fuseaction#"</FONT></B> that I don't have 

a handler for.

      </cfdefaultcase>
I mentioned earlier that <cfdefaultcase> is a very useful tool for debugging; it will alert you if your fusebox is getting a fuseaction that you weren't expecting. This helps eliminate those little errors where you typed showProductInfo instead of displayProductInfo, or editUsrInfo when you meant editUserInfo. At production, you may want to eliminate this, as all errors should be handled by that point. 

Just a few last remarks and then I offer a complete (albeit small) Fusebox program on my web site at http://www.halhelms.com/onlineCourses/BasicFusebox/SimpleContactManagerFB.zip that you can download and examine. 

I try to avoid hardcoding values that can change--most programmers would agree that's just good practice. One of the things that can change is the name of the fusebox file itself. In all of the examples, I've been hardcoding index.cfm. But you may be writing in an environment where some other page is served up by default--perhaps default.cfm. 

Rather than going back and doing global search and replace (the very thought of which chills my cowardly soul), I use the variable, self, as an alias for the fusebox. This value is set in myGlobals.cfm, making for links that look like this: 

<a href="#self#?fuseaction=#attributes.XFA.someFuseaction#">Click me</a>
On another topic, I recommend that you take a look at the specification for Fusedoc. This is really a hybrid "documentation-program definition language" that has really revolutionized the way I write code. There's a free lesson on Fusedocs, so whatcha waitin' for? 

The code for this program, SimpleContactManager, is meant to help you understand Fusebox architecture. It's certainly not production-ready code--for one thing, there's no exception handling--but I wanted to keep it as lean as possible so that you can quickly work your way through the code to the underlying architecture. 

You can set this code to run on your own system by creating an ODBC datasource called "SimpleContactManager" and pointing it at the Access 2000 DB included in the zip file. Click here to download the zipped Fusebox application. 

